

# Seminario de Composición Asistida y Procesamiento de Sonido y Música en Tiempo Real

## *Trabajo Final*

Docente: Dr. Pablo Cetta

Alumno: Lic. Lucas Samaruga

### **Introducción (Notas preliminares)**

La pieza presentada como trabajo final del seminario es para flauta, piano automático y procesamiento en tiempo real empleando el entorno *SuperCollider 3*. La configuración escénica es básica y la parte electroacústica es estéreo, como se detalla en la partitura. Idealmente, el piano automático es un piano acústico que puede ejecutar órdenes MIDI en tiempo real y del cual se capta el sonido producido, mediante uno o dos micrófonos, para luego ser procesado. De no disponerse de un piano así, la pieza puede ser ejecutada con un piano eléctrico, pero debe tenerse en cuenta que el sonido del piano debe ser una fuente sobre el escenario no incluida en la salida estéreo de la parte electroacústica, y su sonido debe ser igualmente captado mediante micrófonos.

En este trabajo, el tiempo es entendido metafóricamente como estático, incluso en lo sucesivo. Para acentuar tal contradicción la música es entendida al mismo tiempo como la superposición de procesos temporales a la manera de los espectralistas. En definitiva, cualquier material musical puede ser objetivable de manera diferente dependiendo de la escala temporal desde la que lo mida (para poder organizarlo compositivamente) y, tal vez, de la atención que se ponga en la escucha. La dialéctica entre objeto y proceso es fundamental para esta pieza.

La indeterminación es otro de los principios importantes de este trabajo, posibilita distintas interpretaciones por parte de los materiales electroacústicos (generados en tiempo real), los cuales pueden variar levemente de ejecución en ejecución. Está presente en varios de los materiales del piano que son generados aleatoriamente a cada ejecución dentro de ciertos límites que permiten mantener su identidad y coherencia formal. Los materiales que consisten en el procesamiento en tiempo real dependen fuertemente de las señales de entrada de la flauta y el piano acústico, el control de la ganancia de los micrófonos puede resultar en diferencias tímbricas más o menos notables en los materiales reactivos (ver Materiales reactivos) y en el contenido de los materiales que emplean buffers para grabar y reproducir los instrumentos acústicos (ver Materiales de resonancia, eco y repetición). Con la indeterminación se pretende que cada ejecución sea única pero que se mantenga la identidad y la coherencia de los materiales como constitutivos de una misma pieza.

La exploración de las proporciones temporales que relacionan la altura con el ritmo es otro de los ejes de esta composición, está basada en las concepciones de Cowell y Stockhausen (ver en Roadas 2002, cap. 2) y es empleada con una finalidad sonora particular como se explicará más adelante. Como nota preliminar respecto de las alturas en sí, es necesario mencionar que tanto las partes instrumentales como las electroacústicas emplean estructuralmente la escala temperada, esto es una decisión previa que surge de la búsqueda de las sonoridades particulares que los *clusters* cromáticos proporcionan.

En cuanto a su contenido, este escrito se centra en el análisis y la exposición de los recursos empleados: las técnicas compositivas (como estructuras previas y/o resultados sonoros buscados), los medios de representación (código de programación, interface gráfica, partitura analógica) y los recursos técnicos de que intervinieron en la composición asistida y el procesamiento de sonido en tiempo real. El trabajo se completa con la partitura para flauta,

el código en *SuperCollider* y la grabación de una ejecución completa de la obra realizada con piano eléctrico y flauta MIDI.

## Notas Sobre la Partitura

La partitura es en realidad una *particella* de flauta que contiene referencias temporales a los materiales desarrollados por el piano y el procesamiento en tiempo real, pero no los representa de manera detallada sino de manera simbólica. La aparición y duración de los materiales son representados generalmente con una punto y una línea (momento de inicio y duración) y complementados con una referencia verbal al material que representan. En dos casos se emplean figuras (símbolos) distintos para diferenciar los materiales de manera más clara. El primero son los rectángulos en la línea del piano en el segundo sistema de la primera hoja, que simbolizan conjuntos de alturas de mayor densidad cronométrica que el resto de los materiales de es instrumento. El segundo caso se da en las secciones H, I y J donde la textura de la parte electroacústica es representada de manera diferente para dar cuenta de una mayor presencia, densidad cronométrica y variaciones tímbricas.

Las letras de ensayo o secciones están rotuladas con letras de la A a la L y determinadas por las necesidades de cambios de la parte electroacústica. Para la ejecución de la obra fue implementada una interface gráfica muy simple que consiste en la presentación de las hojas de la partitura sobre las cuales se puede clicar (dentro de cada sistema y sección) para disparar las distintas partes y seguir la partitura al mismo tiempo.

La parte de flauta propiamente dicha, está escrita en pentagrama, tanto en notación temporal (indicando las duraciones aproximadas de conjuntos de eventos) como en notación proporcional. Las partes de piano y electroacústica están codificadas de manera unívoca en el lenguaje de programación de *SuperCollider* (*sclang*), el cual actúa como la partitura de estos materiales, analizando el código de programación es posible reconstruir la obra incluso con otros recursos.

## El Ritmo y las Duraciones

### *Pulsaciones y octavas rítmicas*

Casi la totalidad de los materiales ejecutados por el piano están compuestos por pulsaciones a distintos *tempi*. Los *tempi* de cada pulsación están en relación directa con la altura empleada. Estos se corresponden directamente con la frecuencia en Hz de las doce notas de la escala temperada en octavas<sup>1</sup> infrasónicas. En el programa, las duraciones se obtiene convirtiendo valores MIDI a frecuencia como se explica en la figura 1.

```
// octava -1 del índice acústico internacional en valores MIDI
// frecuencias de las pulsaciones
(0..11).midicps;
→ [ 8.1757989156437, 8.6619572180273, 9.177023997419, 9.722718241315,
10.300861153527, 10.913382232281, 11.562325709739, 12.24985737443, 12.978271799373,
13.75, 14.56761754744, 15.433853164254 ]

// duraciones de las pulsaciones
(0..11).midicps.reciprocal;
→ [ 0.12231220585509, 0.11544734923405, 0.10896778740921, 0.10285189544532,
0.097079262121457, 0.091630621810536, 0.086487790182016, 0.081633603513409,
```

---

<sup>1</sup>Cuando se hace referencia a la octava de una nota, ya sea mediante el nombre de la nota y un número, por ejemplo “do4”, o solo aludiendo al número de octava, es empleado el índice acústico internacional.

```

0.077051861407949, 0.072727272727273, 0.068645404558669, 0.06479263404657 ]

// octava -2 del índice acústico internacional
// como los números son negativos la escala cromática de do a si está invertida
-1.midicps; // si -2
-12.midicps; // do -2

// frecuencias de las pulsaciones
(-1..-12).midicps;
→ [ 7.7169265821269, 7.2838087737202, 6.875, 6.4891358996866, 6.1249286872148,
5.7811628548693, 5.4566911161407, 5.1504305767636, 4.8613591206575,
4.5885119987095, 4.3309786090136, 4.0878994578219 ]

// duraciones de las pulsaciones
(-1..-12).midicps.reciprocal;
→ [ 0.12958526809314, 0.13729080911734, 0.14545454545455, 0.1541037228159,
0.16326720702682, 0.17297558036403, 0.18326124362107, 0.19415852424291,
0.20570379089063, 0.21793557481843, 0.2308946984681, 0.24462441171017 ]

```

*figura 1*

Al empleo de diferentes octavas infrasónicas para generar una escala de duraciones se lo determina *octava rítmica*. Las diferentes octavas rítmicas generan meramente un conjunto de pulsaciones, ligeramente diferentes, más rápido o más lento.

### *Pulsaciones encadenadas*

Algunos materiales rítmicos están generados por una sucesión de alturas temperadas de las cuales se deriva la duración según la octavas rítmica. Así, por ejemplo, a la sucesión de alturas MIDI 60, 71, 61 y 70 (do<sub>4</sub>, si<sub>4</sub>, reb<sub>4</sub>, sib<sub>4</sub>) le corresponden las duraciones 0.24462441171017, 0.12958526809314, 0.2308946984681, 0.13729080911734 si elegimos la octava rítmica -2 como se muestra en el ejemplo de la figura 2.

```

(
var notasMIDI = [60, 71, 61, 70];
var duraciones = (-12 + (notasMIDI % 12)).midicps.reciprocal;

notasMIDI.postln;
duraciones.postln;

Pbind(
  \midinote, Pseq(notasMIDI, 3),
  \dur, Pseq(duraciones, 3)
).play;
)

```

*figura 2*

La combinación de distintas octavas rítmicas no se emplea en los materiales generados para esta pieza de manera directa, porque los materiales musicales pretendidos no lo requerían. Pero este procedimiento permite organizar las alturas y las duraciones por octavas y octavas rítmicas. En esta obra en particular no se pretende ningún tipo de serialización de las alturas y las duraciones (cosa que el procedimiento en sí no implica), sino la superposición o sucesión de pulsaciones ligeramente diferentes. Por otra parte, la octava de las alturas de las notas es independiente de la octava rítmica y están organizadas según criterios de registración fija, empleados para generar clusters con distintas sonoridades según

la disposición de cada una de las doce alturas temperadas en diferentes registros. En otras palabras, el grado cromático determina la duración de la nota pero el tratamiento de las octavas, tanto rítmicas como “melódicas”, es independiente.

## Los Materiales

Son entendidos como bloques o conjuntos sonoros, pueden definirse por los materiales explicados a continuación, o por una combinación de estos en el caso de los materiales que reaccionan de manera orgánica entre sí. El material sonoro es entendido, en este caso, a nivel de textura musical, tanto verticalmente como linealmente. Esto se ve reflejado en el empleo de la registración fija como el procedimiento que define las altura en la mayoría de los materiales de la pieza.

### *Materiales de piano*

Los materiales que desarrolla el piano (excepto en las secciones G y J donde está subordinado la melodía de la flauta), se generar a partir de un procedimiento algorítmico relativamente sencillo, pero que está diseñado para producir cierto grado de diversidad según el ajuste de determinados parámetros. Estos surgen de la función *~materialPiano* (figura 3).

```
~materialPiano = { arg octr = 0, offr = 62, reg = {60}!12,
                  repp = [1, 2, 3, 4, 5], rept = 4,
                  conjs = [[1, 2, 3], [4, 5, 6], [7, 8, 9], [10, 11, 0]],
                  midiout, legato = 0.9, chan = 0, amp = 0.5;

  var peri = ((-60..-49) + octr).collect({ arg i; i.midicps.reciprocal });
  var offset = peri - (offr % peri);

  Post << "offr = " << offr << "\n"
        << "reg = " << reg << "\n";

  Ptpar(
    conjs.flop.collect({ arg i, j;
      var ad = i.collect({ arg i;
        var r = repp.choose; [Pn(i + reg[i], r), Pn(peri[i], r)]
      }).flop;
    [
      offset[i[0]],
      Pbind(
        \midinote, Pseq(ad[0], rept),
        \dur, Pseq(ad[1], rept),
        \amp, amp,
        \legato, legato,
        \type, \midi,
        \midiout, midiout,
        \chan, chan
      )
    ]
  );
  }).flat;
};
```

*figura 3*

La función *~materialPiano* recibe ciertos parámetros y devuelve una o varias sucesiones de alturas superpuestas. Los parámetros que recibe son: la octava rítmica en

valores midi y relativa a la octava midi -60 (*octr*), el *offset* rítmico (*offr*), un *array* con la registración de cada una de las doce notas, un *array* con las posibles repeticiones por nota (*repp*), las veces que se repite en total el material generado (*rept*), un *array* de dos dimensiones que define la sucesión de acordes (*conjs*), la duración real de cada evento proporcional a la duración entre eventos sucesivos (*legato*) y la amplitud de los eventos (*amp*). Los parámetros *midiout* y *chan* no afectan al material musical, son necesarios para generar las señales midi.

Cada una de las sucesiones de alturas devueltas está constituida por una pulsación (nota repetida) o varias pulsaciones encadenadas (ver el apartado El Ritmo y las Duraciones más arriba), las cuales son definidas por los parámetros *octr*, *repp*, *rept* y por la sucesión de acordes que se pasa como parámetro (*conjs*).

La octava rítmica (*octr*) actúa como el tempo del material generado pero cambia notablemente su carácter, a nivel cualitativo, según sea lento, moderado o rápido. El material como proceso va incrementando su densidad cronométrica (con algunas anticipaciones interpoladas) hasta llegar a la parte de piano solo en la sección F de la partitura.

El *offset* rítmico define el momento a partir del cual comienzan las pulsaciones en relación la polirítmia que generan los doce *tempi* superpuestos comenzando de manera sincronizada en el segundo cero. Esto básicamente define el orden y el momento en el cual aparecen las pulsaciones cuando comienza el material.

### *Materiales de flauta*

Los materiales de la flauta se pueden dividir, de manera general, en tres tipo. El primero son las sucesiones de alturas en las secciones A, C, E y L de la partitura, las cuales se caracterizan por ser pulsaciones o duraciones largas iguales o aproximadamente iguales. La escritura no proporcional de las secciones A y L acepta una interpretación indeterminada en cuanto a su duración real dentro de las duraciones indicadas para los conjuntos de eventos. El segundo tipo de material de la flauta es el de las secciones G y J. Este consiste claramente en una melodía compuesta motivica y formalmente de manera tradicional. El último tipo es el material de la sección I de la partitura, que está compuesto algorítmicamente y emplea procesos estocásticos. El algoritmo que lo determina puede verse en la figura 4.

```
thisThread.randSeed = 3008;

// 1' a 0.8
~dur = 80;
// 0.4 / (80 * 4) (0.1 a 0.5 en 80 pasos)
~wcDur = Pseries(0.1, 0.00125, inf).asStream;
// 0.56 / (80 * 4) (0.1 a 0.66 en 80 pasos)
~wcOct = Pseries(0.1, 0.00175, inf).asStream;
~transp = 74;

p = Pbind(
  \midinote,
    Pxrand([0, 1, 2, 3], ~dur * 4) + ~transp
  +
  Pfunc({
    var next = ~wcOct.next;
    [0, 14, -13].wchoose([1 - next, next/2, next/2])
  }),
  \dur, Pfunc({
    var next = ~wcDur.next;
    [0.2, 0.3].wchoose([1 - next, next]) * 0.8
  })
)
```

);

figura 4

Está compuesto utilizando los recursos de la librería de *patterns* de SuperCollider 3. Nótese que los *patterns* son entendidos como sucesiones de valores o funciones sobre las cuales se puede operar matemáticamente. Para esta melodía en particular se definen solo el comportamiento rítmico y el comportamiento de las alturas según funciones probabilidad tanto homogénea (*Pxrand*) como sopesada (*wchoose*) y la variación direccional de las probabilidades sopesadas (*~wcDur* y *~wcOct*).

El ritmo está constituido por dos valores relacionados proporcionalmente como 1 y 2/3 (0.3 y 0.2 respectivamente) los cuales tiene una probabilidad de aparición diferente que cambia linealmente a lo largo del pasaje. Los coeficientes para la probabilidad del valor proporcional 2/3 van de 0.9 a 0.5 y los del valor 1 van de 0.1 a 0.5.

Las alturas están definidas por un *cluster* cromático de cuatro notas a partir del re5 del cual se elige aleatoriamente con probabilidad homogénea una nota a la vez pero sin repetir la misma nota dos veces seguidas (función *Pxrand*). Al resultado de esta función se le suma el resultado de la función *Pfunc*, que consiste en seleccionar aleatoriamente la no transposición, la transposición ascendente de octava mas tono o la transposición descendente de octava más semitono. Los coeficientes de las probabilidades van de 0.9 a 0.44 para la no transposición y de 0.05 a 0.33 para cada una de las transposiciones. El salto de octava transpuesto define *clusters* cromáticos de cuatro notas una octava mas un tono hacia arriba y una octava mas un semitono hacia abajo.

La melodía resultante son permutaciones del *cluster* cromático central en valores de corchea que progresivamente va incrementando la aparición de valores de corchea con punto y los saltos de registro más transposición, generando una inestabilidad rítmica progresivamente más notable y contrapunto lineal debido a los saltos de octava.

### *Materiales reactivos*

La pieza se compone de dos materiales que reaccionan mediante la detección de la altura a lo que toca la flauta. Uno acústico, entre la flauta y el piano automático y otro electroacústico, entre la flauta y la electroacústica en tiempo real.

El material que se genera en el piano en respuesta a la melodía de la flauta, actúa como complemento tímbrico y/o textural de los materiales de las secciones G y J de la flauta. Es en rigor, parte de un solo material compuesto pero está expuesto por separado a los fines del análisis de los recursos empleados. La definición de instrumento (en adelante DI) *\cpsmidiDet* (figura 5), realiza la detección de la frecuencia de la flauta, mediante la unidad generadora (en adelante UG) *Pitch* que luego es convertida a valores MIDI y enviada nuevamente al lenguaje para ser procesada mediante el objeto *OSCresponder* (figura 6).

```
SynthDef(\cpsmidiDet, { arg in;
  var freq, hasFreq, dust;
  #freq, hasFreq = Pitch.kr(In.ar(in));
  dust = Dust.kr(20) * EnvGen.kr(Env.asr(0.01, 1, 0.01), hasFreq);
  SendTrig.kr(dust, 0, freq.cpsmidi + TRand.kr(-1, 1, dust));
}).add;
```

figura 5

Lo que realiza el objeto *OSCresponder* es generar los eventos midi y enviarlos al piano. El material consiste en la generación aleatoria de varias notas un semitono arriba o abajo de la altura que está tocando la flauta, con intensidad variable y un ritmo aleatorio de

aproximadamente 20 ataques por segundo.

```
~gflAmp = 0.1;
~gflAmpFactor = 8;

~flResponder = OSCresponder(s.addr, \tr, { arg time, resp, msg;
  ~flAmpBus.get({arg i; ~gflAmp = i});
  (
    midinote: msg.last.round,
    dur: 0.1,
    amp: rrand(0.01, ~gflAmp * ~gflAmpFactor),
    type: \midi,
    midiout: ~midiOut
  ).play;
});
```

figura 6

El material electroacústico que responde a la frecuencia de la flauta, está compuesto de varias partes y estratos. El material solo, cuando no responde a ninguna frecuencia como en la sección H, está compuesto por un *cluster* móvil que emplea dos fuentes distintas: los *buffers* de repetición filtrados y la síntesis granular en base al contenido de los mismos *buffers* (figuras 7 y 8) (ver más abajo Materiales de resonancia, eco y repetición).

```
SynthDef(\reactGrain, { arg out = 0, freq = 440, amp = 0.0005, buffer,
  kFreq, kAmp, banda = 0.5;

  var fgate = (
    (In.kr(kFreq) > (freq * (2 ** ((banda * -1)/12))))
    * (In.kr(kFreq) < (freq * (2 ** (banda/12))))
    * In.kr(kFreq, 2)[1]
  ).lag(1);

  var grains = TGrains.ar(
    numChannels: 2,
    trigger: Impulse.ar(freq),
    bufnum: buffer,
    rate: 1,
    centerPos: Rand(0, BufDur.kr(buffer))
      * LFNoise2.ar(0.1 + (kAmp * fgate), 0.1, 0.1),
    dur: 1.25 / freq,
    pan: -1
  );

  var env = EnvGen.kr(Env.perc(2.5, 0.5), doneAction:2);
  amp = amp * (1 + (5 * fgate));

  Out.ar(out, grains * env * amp);
}).add;
```

figura 7

```
alte.do { arg i;
  repe = [3, 4, 5, 6].choose.postln;
  p = Ptpar([
    { rrand(0.2, repe) }!dens,
    { Pbind(
      \instrument, \reactGrain,
```

```

        \group, ~detGroup2,
        \out, Prand([0, 1], repe),
        \midinote, Pxrand((48..96), repe),
        \amp, ~ampReactGrain,
        \buffer, ~supraBuffers.choose,
        \kFreq, ~flFreqBus,
        \kAmp, ~flAmpBus,
        \banda, 2,
        \dur, Pxrand([0.5, 1.0, 1.5] * rrand(1.0, 2.0), repe)
    })!dens
].flop.flat).play;
(repe * 1.5).wait;

repe = [5, 6, 7, 8].choose.postln;
p = Ptpar([
    { rrand(0.2, repe) }!dens2,
    { Pbind(
        \instrument, \reactFil,
        \group, ~detGroup2,
        \out, Prand([0, 1], repe),
        \midinote, Pxrand((36..108), repe),
        \amp, ~ampReactFil,
        \buffer, ~supraBuffers.choose,
        \kFreq, ~flFreqBus,
        \kAmp, ~flAmpBus,
        \banda, 4,
        \dur, Pxrand([0.5, 1.0, 1.5] * rrand(1.0, 2.0), repe)
    })!dens2
].flop.flat).play;
(repe * 1.5).wait;
};

```

*figura 8*

Cuando el material no aparece solo, los ataques que se corresponden con la frecuencia que está tocando la flauta sobresalen del continuo generando un nuevo estrato que puede fusionarse con el timbre de la flauta lo que se produce de manera clara en la sección final de la pieza. Los ataques sobresalen del continuo porque su intensidad se incrementa y en el caso de la síntesis granular también se produce un cambio tímbrico cuya magnitud está en relación a la amplitud de la nota ejecutada por la flauta.

### *Materiales de resonancias, eco y repetición*

Otra parte del procesamiento en tiempo real consiste en la grabación del sonido del piano y la flauta en tres conjuntos de *buffers* de manera circular. Estos son reproducidos, durante toda la obra al mismo tiempo que graban y generan tres estratos distintos según las duraciones de los conjuntos sean mayores o menores. Las duraciones de los conjuntos de *buffers* están determinadas por octavas rítmicas y son las siguientes,

resonancia:

```

~micro = (12..23).midicps.reciprocal;
→ [ 0.061156102927543, 0.057723674617025, 0.054483893704607, 0.051425947722658,
0.048539631060728, 0.045815310905268, 0.043243895091008, 0.040816801756704,
0.038525930703974, 0.036363636363636, 0.034322702279334, 0.032396317023285 ]

```



eco:

```
~meso = (-25..-36).midicps.reciprocal;  
→ [ 0.51834107237256, 0.54916323646935, 0.58181818181818, 0.61641489126359,  
0.65306882810727, 0.69190232145613, 0.73304497448429, 0.77663409697166,  
0.82281516356253, 0.8717422992737, 0.92357879387241, 0.97849764684069 ]
```

repetición:

```
~supra = (-73..-84).midicps.reciprocal;  
→ [ 8.293457157961, 8.7866117835096, 9.3090909090909, 9.8626382602174,  
10.449101249716, 11.070437143298, 11.728719591749, 12.426145551546, 13.165042617,  
13.947876788379, 14.777260701959, 15.655962349451 ]
```

Como su nombre lo indica, estos tres conjuntos de buffers producen efectos de resonancia (*~micro*), eco (*~meso*) y repetición (*~supra*), con la salvedad de que lo que en realidad graban los *buffers* de repetición es la mezcla de la salida de los *buffers* de eco y no el sonido directo. La presencia o la ausencia (la intensidad, perceptible o no) de los *buffers* de resonancia, genera un cambio tímbrico y espacial entre el sonido directo de los instrumentos y la parte electroacústica si se respeta la disposición escénica.

La grabación circular la posibilita la UG *RecordBuf* que permite grabar en loop y controlar la ganancia de la señal de entrada y la del contenido previo del buffer, como puede verse en la DI *recIn* en la figura 9.

```
SynthDef(\recIn, { arg inBus, buffer, umbral = 0, recLevel = 0,  
lagRecLevel = 0.1, preLevel = 0.9;  
var input = In.ar(inBus);  
var ampGate = Amplitude.ar(input) > umbral;  
input = input * ampGate.lag(0.2);  
RecordBuf.ar(  
input, buffer,  
recLevel:recLevel.lag(lagRecLevel),  
preLevel:preLevel, loop: 1  
);  
}).add;
```

figura 9

### Las densidades y distribuciones de los clusters

Las alturas de la pieza varían según varios factores que fueron definidos manualmente, sin la intervención de procesos automáticos. Analizar detalladamente el comportamiento de los *clusters* sería demasiado extenso (y probablemente no relevante) para este análisis, en su lugar se exponen los principios que se fueron tenidos en cuenta.

Los clusters varían a lo largo de la pieza en densidad armónica (cantidad de notas superpuestas) y densidad cromática (cantidad de alturas distintas). Las pulsaciones encadenadas, que caracterizan los materiales del piano, varían entre lo melódico y lo armónico según la octava rítmica que se emplee. Las octavas rítmicas cercanas al cero o negativas (pasadas como parámetro a la función *~materialPiano*) generan comportamientos melódicos, por la baja densidad cronométrica, y las positivas generan una mayor densidad cronométrica y la superposición. A su vez el parámetro de la altura está constituido por una sucesión de acordes que varía su densidad cromática entre una y cuatro alturas.

La parte electroacústica del material reactivo entre flauta y electroacústica, es un

cluster móvil constante en cuanto a densidad cromática pero que varía sutilmente en cuanto a la densidad armónica. Ésta no es siempre homogénea en todo el registro y tiene un comportamiento simétrico entre densidad homogénea en la sección H de la partitura, densidad centrada en el *cluster* cromático de la flauta en la sección I y densidad homogénea nuevamente en la sección final.

## Principio Formal

La organización meso y macro formal está dada por la utilización recursiva de la simetría ABA, por ejemplo, la parte de piano solo, en la sección F de la partitura, está compuesta por dos materiales distintos y sus variaciones y organizada formalmente como se muestra en la figura 10. Las sub secciones de la parte B no son estrictamente simétricas pero pueden ser entendidas como la imbricación de las simetrías b1 b2 b1 y b2 b1 b2.




figura 10

La simetría está determinada por ciertas cualidades de los materiales musicales pero sin correlato con las duraciones de estos. El procedimiento no se aplica de manera absoluta, solo sirve como principio estructurante para la determinación de la forma.

El resto de la pieza está organizado en base al mismo principio pero de manera más difuminada, ya sea por procesos lineales superpuestos o por la no proporcionalidad en las duraciones de las secciones.

## Organización Formal

La pieza está organizada en secciones (letras de ensayo) según las necesidades de cambio de la parte electroacústica y no en relación directa con los principios formales. En este apartado se analiza la estructura formal según fue concebida a nivel macro formal, tomando como referencia las letras de ensayo (secciones) de la partitura.

Simetría 1, letras de ensayo A y B: Conformen la primera simetría formal entre la parte de piano lento y la parte de flauta y la reaparición del material del piano. Esto puede verse claramente en el primer sistema de la partitura. Los materiales de resonancia, eco y repetición no responden a este comportamiento simétrico sino que generan variaciones en relación a los materiales del piano y la flauta.

Simetría 2, letras de ensayo C, D y E: Conforman la segunda simetría entre el material de la flauta, el material del piano y la repetición de la flauta. En este sentido es una inversión del comportamiento de la primera sección. Sin embargo la sección del piano está subdividida en dos partes, compuestas por la aparición del material denominado “piano rápido”, claramente simétrico y representado como rectángulos en la partitura, y la reaparición del material “piano lento”.

La concatenación de las simetrías 1 y 2 pueden ser entendidas estrictamente como la alternancia AB entre las apariciones variadas del piano y la flauta, esto es porque el principio formal es ambiguo en este caso y no necesariamente representa la forma resultante de manera unívoca. Esto no es un hecho indeseable sino todo lo contrario, el principio no necesariamente determina la forma resultante e incluso, recuérdese, ningún procedimiento empleado de manera estricta.

Simetría 3, letra de ensayo F: Fue concebida como una pequeña pieza para piano solo dentro de la totalidad de la obra. Se basa en los mismos materiales generados por la

función *~materialPiano* y se organiza como fue explicado en el apartado “Principio Formal”. Sin embargo esta sección introduce, dentro de la parte central de su simetría, el tremolo de nota repetida que genera un sonido continuo, iterado, y el *clusters* de *crescendi* de notas tremoladas, con distintas envolventes dinámicas, que genera una sonoridad poco usual como material para piano solo. Este material es simplemente una variación de los materiales de octavas rítmicas generados por la función *~materialPiano*.

Simetría 4, letras de ensayo G, H, I, y J: Aquí el procedimiento de simetría se da de manera clara entre el material de flauta y piano reactivo y el material de flauta y electroacústica reactiva. Temporalmente, las duraciones de las secciones que constituyen la simetría son mucho mayores a las de las simetrías anteriores.

Simetría 5, letras de ensayo I, J, K y L: Por último, en esta sección formal puede entenderse un procedimiento simétrico con respecto a las apariciones de la electroacústica reactiva de las letras H, I y L y el material de piano de la letra J. Se produce nuevamente una imbricación con la simetría anterior que rompe con el principio formal. Además reaparece, se interpola, una variación del material “piano lento” del principio de la obra en la letra de ensayo K, el cual tiene la intención de actuar a la manera de una coda junto con el material de flauta y electroacústica reactiva de la letra L.

## **Bibliografía**

CETTA, P., 2004 “Procesamiento en tiempo real en la obra de Luigi Nono”. En *Altura – Timbre – Espacio*. Cuaderno No 5 del Instituto de Investigación Musicológica “Carlos Vega”, pp. 235-257. EDUCA. Buenos Aires.

DI LISCIA, Oscar. P., 2004 Generación y procesamiento de sonido y música a través del programa Csound. Bernal: Universidad Nacional de Quilmes.

PUCKETTE, M., 2007 The Theory and Technique of Electronic Music. : World Scientific Publishing Co. Pte. Ltd.

ROADS, C., 2002 Microsound. Cambridge y Massachusetts: The MIT press.

ROADS, C., 2000 The Computer Music Tutorial. Cambridge y Massachusetts: The MIT press.

Documentación de SuperCollider, en <http://supercollider.sourceforge.net> (ultimo acceso 2 de noviembre de 2010)